
LISREL and Mplus code to accompany: 

curve models. Invited talk at the Advances in Longitudinal Methods in the Social and Behavioral 

Sciences conference, The Center for Integrated Latent Variable Research, College Park, MD. 

 

Preacher, K. J., & Hancock, G. R. (2012). On interpretable reparameterizations of linear and nonlinear 

latent growth curve models. In Harring, J. R., & Hancock, G. R. (Eds.), Advances in longitudinal methods 

in the social and behavioral sciences (pp. 25-58). Charlotte, NC: Information Age Publishing. 

 

Three ways to estimate an aperture parameter 

 

Method 1: Lambda shift method 

 
LATENT GROWTH CURVE OF HANCOCK & CHOI DATA, EX. 2 
DA NI=4 NO=198 MA=CM 
CM 
11.000 
5.860 13.000 
6.205 8.094 14.000 
6.103 8.798 10.177 16.000 
ME 
3.3 3.7 4.0 4.2 
MO NY=4 NE=2 LY=FU,FI BE=FU,FI TY=FI AL=FR PS=SY,FI TE=DI,FR AP=1 
LE 
INT SLP 
FR PS 1 1 PS 2 2 !PS 2 1 
VA 1 LY 1 1 LY 2 1 LY 3 1 LY 4 1 
CO LY 1 2 = 4 - PA 1 
CO LY 2 2 = 6 - PA 1 
CO LY 3 2 = 7 - PA 1 
CO LY 4 2 = 8 - PA 1 
PD 
OU IT=5000 AD=OFF ND=4 

 

Method 2: Phantom variable approach 

 
LATENT GROWTH CURVE OF HANCOCK & CHOI DATA, EX. 2 
DA NI=4 NO=198 MA=CM 
CM 
11.000 
5.860 13.000 
6.205 8.094 14.000 
6.103 8.798 10.177 16.000 
ME 
3.3 3.7 4.0 4.2 
MO NY=4 NE=3 LY=FU,FI BE=FU,FI TY=FI AL=FI PS=SY,FI TE=DI,FR 
LE 
INT SLP PHANTOM 
FR AL 1 AL 2 
PA PS 
 1 
 0 1 



 0 0 0 
MA PS 
 .5 
  0 .5 
  0 0 0 
PA LY 
0 0 0 
0 0 0 
0 0 0 
0 0 0 
MA LY 
1 4 -1 
1 6 -1 
1 7 -1 
1 8 -1 
PA BE 
 0 0 0 
 0 0 0 
 0 1 0 
MA BE 
 0 0 0 
 0 0 0 
 0 -4.3 0 
PD 
OU IT=5000 AD=OFF ND=4 

 

Method 3: Structured latent curve approach 

 
LATENT GROWTH CURVE OF HANCOCK & CHOI DATA, EX. 2 
DA NI=4 NO=198 MA=CM 
CM 
11.000 
5.860 13.000 
6.205 8.094 14.000 
6.103 8.798 10.177 16.000 
ME 
3.3 3.7 4.0 4.2 
MO NY=4 NE=3 LY=FU,FI TY=FI AL=FR PS=SY,FI TE=DI,FR AP=1 
LE 
INT SLP APERTURE 
PA PS 
 1 
 0 1 
 0 0 0 
MA PS 
 .5 
  0 .5 
  0  0  0 
PA LY 
 0 0 0 
 0 0 0 
 0 0 0 
 0 0 0 
MA LY 
 1 0 0 
 1 0 0 



 1 0 0 
 1 0 0 
CO LY 1 2 = 4 - PA 1 
CO LY 2 2 = 6 - PA 1 
CO LY 3 2 = 7 - PA 1 
CO LY 4 2 = 8 - PA 1 
CO LY 1 3 = -1*AL(2) 
CO LY 2 3 = -1*AL(2) 
CO LY 3 3 = -1*AL(2) 
CO LY 4 3 = -1*AL(2) 
CO AL 3 = PA 1 
PD 
OU IT=5000 AD=OFF ND=4 

 

A reparameterized Gompertz structured latent growth curve model with random coefficients 

 

Example application to ECLS-K mathematics data (kindergarten through 8
th

 grade). Random coefficients 

represent the upper asymptote, surge point, and surge slope (see Choi, Hancock, & Harring, 2009) for 

definitions of these terms. 

 

Choi, J., Harring, J. R., & Hancock, G. R. (2009). Latent growth modeling for logistic response functions. 

Multivariate Behavioral Research, 44, 620-645. 

 
TITLE: ECLS-K math, reparameterized Gompertz curve; 

DATA: FILE = eclsk.dat; 

VARIABLE: NAMES = id !gender !0=boy, 1=girl; uncomment gender for conditional model 

m_fk m_sk m_f1 m_s1 m_s3 m_s5 m_s8; 

USEVARIABLES ARE gender m_fk-m_s8; 

MISSING = .; 

ANALYSIS: 

ALGORITHM = INTEGRATION; 

INTEGRATION = MONTECARLO; 

MODEL: 

 

!Factor Loadings 

g0 BY m_fk*(Lg01) 

m_sk-m_s8(Lg02-Lg07); 

t0 BY m_fk*(Lt01) 

m_sk-m_s8(Lt02-Lt07); 

g3 BY m_fk*(Lg31) 

m_sk-m_s8(Lg32-Lg37); 

 

!Means 

[m_fk-m_s8@0]; [g0*147](mu_g0); [t0@0]; [g3*22](mu_g3); 

 

!Variances and covariances 

m_fk-m_s8; g0*495 t0*.24 g3; g0 WITH t0 g3; t0 WITH g3; 

 

!Regressions 

!g0 t0 g3 ON gender; !uncomment line for conditional model 

 

MODEL CONSTRAINT: 

NEW(mu_t0*.7); !Introduce mean of surge point 

 



!Asymptote loadings 

Lg01 = exp(-1*exp(((mu_g3*exp(1)*(mu_t0-0.0))/mu_g0))) 

+((mu_g3*exp(1)*(mu_t0-0.0)*exp(mu_g3*exp(1)*(mu_t0-0.0)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-0.0)/mu_g0)))/mu_g0); 

Lg02 = exp(-1*exp(((mu_g3*exp(1)*(mu_t0-0.5))/mu_g0))) 

+((mu_g3*exp(1)*(mu_t0-0.5)*exp(mu_g3*exp(1)*(mu_t0-0.5)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-0.5)/mu_g0)))/mu_g0); 

Lg03 = exp(-1*exp(((mu_g3*exp(1)*(mu_t0-1.0))/mu_g0))) 

+((mu_g3*exp(1)*(mu_t0-1.0)*exp(mu_g3*exp(1)*(mu_t0-1.0)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-1.0)/mu_g0)))/mu_g0); 

Lg04 = exp(-1*exp(((mu_g3*exp(1)*(mu_t0-1.5))/mu_g0))) 

+((mu_g3*exp(1)*(mu_t0-1.5)*exp(mu_g3*exp(1)*(mu_t0-1.5)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-1.5)/mu_g0)))/mu_g0); 

Lg05 = exp(-1*exp(((mu_g3*exp(1)*(mu_t0-3.5))/mu_g0))) 

+((mu_g3*exp(1)*(mu_t0-3.5)*exp(mu_g3*exp(1)*(mu_t0-3.5)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-3.5)/mu_g0)))/mu_g0); 

Lg06 = exp(-1*exp(((mu_g3*exp(1)*(mu_t0-5.5))/mu_g0))) 

+((mu_g3*exp(1)*(mu_t0-5.5)*exp(mu_g3*exp(1)*(mu_t0-5.5)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-5.5)/mu_g0)))/mu_g0); 

Lg07 = exp(-1*exp(((mu_g3*exp(1)*(mu_t0-8.5))/mu_g0))) 

+((mu_g3*exp(1)*(mu_t0-8.5)*exp(mu_g3*exp(1)*(mu_t0-8.5)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-8.5)/mu_g0)))/mu_g0); 

 

!Surge point loadings 

Lt01 = -1*mu_g3*exp(1)*exp(mu_g3*exp(1)*(mu_t0-0.0)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-0.0)/mu_g0)); 

Lt02 = -1*mu_g3*exp(1)*exp(mu_g3*exp(1)*(mu_t0-0.5)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-0.5)/mu_g0)); 

Lt03 = -1*mu_g3*exp(1)*exp(mu_g3*exp(1)*(mu_t0-1.0)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-1.0)/mu_g0)); 

Lt04 = -1*mu_g3*exp(1)*exp(mu_g3*exp(1)*(mu_t0-1.5)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-1.5)/mu_g0)); 

Lt05 = -1*mu_g3*exp(1)*exp(mu_g3*exp(1)*(mu_t0-3.5)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-3.5)/mu_g0)); 

Lt06 = -1*mu_g3*exp(1)*exp(mu_g3*exp(1)*(mu_t0-5.5)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-5.5)/mu_g0)); 

Lt07 = -1*mu_g3*exp(1)*exp(mu_g3*exp(1)*(mu_t0-8.5)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-8.5)/mu_g0)); 

 

!Surge slope loadings 

Lg31 = -1*exp(1)*(mu_t0-0.0)*exp(mu_g3*exp(1)*(mu_t0-0.0)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-0.0)/mu_g0)); 

Lg32 = -1*exp(1)*(mu_t0-0.5)*exp(mu_g3*exp(1)*(mu_t0-0.5)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-0.5)/mu_g0)); 

Lg33 = -1*exp(1)*(mu_t0-1.0)*exp(mu_g3*exp(1)*(mu_t0-1.0)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-1.0)/mu_g0)); 

Lg34 = -1*exp(1)*(mu_t0-1.5)*exp(mu_g3*exp(1)*(mu_t0-1.5)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-1.5)/mu_g0)); 

Lg35 = -1*exp(1)*(mu_t0-3.5)*exp(mu_g3*exp(1)*(mu_t0-3.5)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-3.5)/mu_g0)); 

Lg36 = -1*exp(1)*(mu_t0-5.5)*exp(mu_g3*exp(1)*(mu_t0-5.5)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-5.5)/mu_g0)); 

Lg37 = -1*exp(1)*(mu_t0-8.5)*exp(mu_g3*exp(1)*(mu_t0-8.5)/mu_g0) 

*exp(-1*exp(mu_g3*exp(1)*(mu_t0-8.5)/mu_g0)); 

 



The Jenss-Bayley model reparameterized to estimate the effect of cumulative breastfeeding on infant 

weight at any desired age 

 

We did not have permission to post the Cebu infant data, but we provide Mplus code below to show how 

to estimate the model. 

 
TITLE: cebu growth data (jenss-bayley) with mobile intercept; 

DATA: FILE IS cebu_wide_more.dat; 

VARIABLE: NAMES ARE id momht rural male age0-age12 br0-br12 h0-h12 w0-w12 b0 b2 b4 b6 

b8 b10 b12 b14 b16 b18 b20 b22 b24 xb6 xb8 xb10 xb12 xb14 xb16 xb18 xb20 xb22 xb24; 

USEVARIABLES ARE cbf !comment out 'cbf' for unconditional model 

x0-x12; !b0-b24 cumulative BF means from age 0m; xb6-xb24 from age 6m 

MISSING ARE ALL (-999); 

USEOBSERVATIONS ARE id NE 1600044; !omit outlier 

DEFINE: !below, choose to model height or weight 

!x0=h0;x1=h1;x2=h2;x3=h3;x4=h4;x5=h5;x6=h6;x7=h7;x8=h8;x9=h9;x10=h10;x11=h11;x12=h12; 

x0=w0;x1=w1;x2=w2;x3=w3;x4=w4;x5=w5;x6=w6;x7=w7;x8=w8;x9=w9;x10=w10;x11=w11;x12=w12; 

!comment out next line for unconditional model 

cbf=b12; ! <- SET AGE FOR CUMULATIVE BREASTFEEDING (mos.); 

ANALYSIS: ESTIMATOR IS ML; ITERATIONS ARE 10000; !BOOTSTRAP IS 300; 

MODEL: [x0-x12@0]; x0-x12*.1(v1); x0-x11 PWITH x1-x12*.027(v2); 

fa*1.2; fb*.005; fc*.288; fd*.02; fa WITH fb fc fd; fb WITH fc fd; fc WITH fd; 

[fa*9](mfa); [fb*.13](mfb); [fc*1.2](mfc); [fd@0]; ! fa, fb, and fc are linear 

fa BY x0-x12@1; 

fb BY x0*(b0); fb BY x1-x12(b1-b12); 

fc BY x0*(c0); fc BY x1-x12(c1-c12); 

fd BY x0*(d0); fd BY x1-x12(d1-d12); 

!comment out next line for unconditional model 

fa ON cbf*-.53; fb ON cbf*.05; fc ON cbf*-.55; fd ON cbf*-.23; 

 

!jenss-bayley with a mobile intercept 

MODEL CONSTRAINT: NEW(mfd*-.334 t0); t0=12; ! <- SET INTERCEPT LOCATION (mos.) 

 

b0=0-t0; c0=exp(mfc+mfd*t0)-exp(mfc+mfd*0); 

b1=2-t0; c1=exp(mfc+mfd*t0)-exp(mfc+mfd*2); 

b2=4-t0; c2=exp(mfc+mfd*t0)-exp(mfc+mfd*4); 

b3=6-t0; c3=exp(mfc+mfd*t0)-exp(mfc+mfd*6); 

b4=8-t0; c4=exp(mfc+mfd*t0)-exp(mfc+mfd*8); 

b5=10-t0; c5=exp(mfc+mfd*t0)-exp(mfc+mfd*10); 

b6=12-t0; c6=exp(mfc+mfd*t0)-exp(mfc+mfd*12); 

b7=14-t0; c7=exp(mfc+mfd*t0)-exp(mfc+mfd*14); 

b8=16-t0; c8=exp(mfc+mfd*t0)-exp(mfc+mfd*16); 

b9=18-t0; c9=exp(mfc+mfd*t0)-exp(mfc+mfd*18); 

b10=20-t0; c10=exp(mfc+mfd*t0)-exp(mfc+mfd*20); 

b11=22-t0; c11=exp(mfc+mfd*t0)-exp(mfc+mfd*22); 

b12=24-t0; c12=exp(mfc+mfd*t0)-exp(mfc+mfd*24); 

 

d0=t0*exp(mfc+mfd*t0)-0*exp(mfc+mfd*0); 

d1=t0*exp(mfc+mfd*t0)-2*exp(mfc+mfd*2); 

d2=t0*exp(mfc+mfd*t0)-4*exp(mfc+mfd*4); 

d3=t0*exp(mfc+mfd*t0)-6*exp(mfc+mfd*6); 

d4=t0*exp(mfc+mfd*t0)-8*exp(mfc+mfd*8); 

d5=t0*exp(mfc+mfd*t0)-10*exp(mfc+mfd*10); 

d6=t0*exp(mfc+mfd*t0)-12*exp(mfc+mfd*12); 

d7=t0*exp(mfc+mfd*t0)-14*exp(mfc+mfd*14); 



d8=t0*exp(mfc+mfd*t0)-16*exp(mfc+mfd*16); 

d9=t0*exp(mfc+mfd*t0)-18*exp(mfc+mfd*18); 

d10=t0*exp(mfc+mfd*t0)-20*exp(mfc+mfd*20); 

d11=t0*exp(mfc+mfd*t0)-22*exp(mfc+mfd*22); 

d12=t0*exp(mfc+mfd*t0)-24*exp(mfc+mfd*24); 

 

OUTPUT: TECH1 TECH3 STDYX; !CINTERVAL(BCBOOTSTRAP); 

 

 


